Example 15 Verifying the BSSN equations

This is short example verifies two of the main equations in the Phys Rev D paper by Miguel Alcubierre, Bernd Brugmann etal. (Phys.Rev.D. (62) 044034 (2000)).

The code for the full set of BSSN equations can be found at `https://github.com/leo-brewin/adm-bssn-equations`

```
{a,b,c,d,e,f,i,j,k,l,m,n,o,p,q,r,s,u#}::Indices(position=independent,values={t,x,y,z}).
{t,x,y,z}::Coordinate.

\partial{#}::PartialDerivative.
D{#}::Derivative.
DBar{#}::Derivative.

N::Depends(t,x,y,z).

g_{a b}::Symmetric.
g^{a b}::Symmetric.
g_{a}^{b}::KroneckerDelta.
g^{a}_{b}::KroneckerDelta.

g_{a b}::Depends(t,x,y,z).
g^{a b}::Depends(t,x,y,z).

gBar_{a b}::Symmetric.
gBar^{a b}::Symmetric.
gBar_{a}^{b}::KroneckerDelta.
gBar^{a}_{b}::KroneckerDelta.

gBar_{a b}::Depends(t,x,y,z).
gBar^{a b}::Depends(t,x,y,z).

trK::LaTeXForm("K").
detg::LaTeXForm("g").
ABar{#}::LaTeXForm("{\bar{A}}").
DBar{#}::LaTeXForm("{\bar{D}}").
```

15.1 Evolution equation for ϕ

```
phi     := \phi -> (1/12) \log(detg).
gdotK   := g^{i j} K_{i j} -> trK.
DdetgDt := \partial_{t}{detg} -> detg g^{i j} \partial_{t}{g_{i j}}.

DgijDt  := \partial_{t}{g_{i j}} -> -2 N K_{i j}.

dlog    := \partial_{a?}{\log(A?)} -> (1/A?)\partial_{a?}{A?}.
dexp    := \partial_{a?}{\exp(A?)} -> \exp(A?)\partial_{a?}{A?}.

dotphi  := \partial_{t}{\phi}.

substitute (dotphi, phi)                # cdb (ex-15-02.101,dotphi)
substitute (dotphi, dlog)               # cdb (ex-15-02.102,dotphi)
substitute (dotphi, DdetgDt)            # cdb (ex-15-02.103,dotphi)
substitute (dotphi, DgijDt)             # cdb (ex-15-02.104,dotphi)
substitute (dotphi, gdotK)              # cdb (ex-15-02.105,dotphi)
map_sympy  (dotphi, "simplify")         # cdb (ex-15-02.106,dotphi)

DphiDt := \partial_{t}{\phi} -> @(dotphi).

checkpoint.append (dotphi)
```

$$\frac{d\phi}{dt} = \frac{1}{12}\partial_t \left(\log\left(g\right)\right) \tag{ex-15-02.101}$$

$$= \frac{1}{12} g^{-1} \partial_t g \tag{ex-15-02.102}$$

$$= \frac{1}{12} g^{-1} g g^{ij} \partial_t g_{ij} \tag{ex-15-02.103}$$

$$= -\frac{1}{6} g^{-1} g g^{ij} N K_{ij} \tag{ex-15-02.104}$$

$$= -\frac{1}{6} g^{-1} g K N \tag{ex-15-02.105}$$

$$= -\frac{1}{6} K N \tag{ex-15-02.106}$$

15.2 Evolution equation for $\bar{g}_{ij}$

```
gBarij := gBar_{i j} -> \exp(-4\phi) g_{i j}.
Kij    := K_{i j} -> A_{i j} + (1/3) g_{i j} trK.
A2ABar := \exp(-4\phi) A_{i j} -> ABar_{i j}.
ABar2A := ABar_{i j} -> \exp(-4\phi) A_{i j}.

dotgBarij := \partial_{t}{gBar_{i j}}.

substitute   (dotgBarij, gBarij)         # cdb (ex-15-03.101,dotgBarij)
product_rule (dotgBarij)                 # cdb (ex-15-03.102,dotgBarij)
substitute   (dotgBarij, dexp)           # cdb (ex-15-03.103,dotgBarij)
substitute   (dotgBarij, DgijDt)         # cdb (ex-15-03.104,dotgBarij)
substitute   (dotgBarij, DphiDt)         # cdb (ex-15-03.105,dotgBarij)
substitute   (dotgBarij, Kij)            # cdb (ex-15-03.106,dotgBarij)
distribute   (dotgBarij)                 # cdb (ex-15-03.107,dotgBarij)
map_sympy    (dotgBarij, "simplify")     # cdb (ex-15-03.108,dotgBarij)
substitute   (dotgBarij, A2ABar)         # cdb (ex-15-03.109,dotgBarij)

DgBarijDt := \partial_{t}{gBar_{i j}} -> @(dotgBarij).

checkpoint.append (dotgBarij)
```

$$\frac{d\bar{g}_{ij}}{dt} = \partial_t \left(\exp\left(-4\phi\right) g_{ij} \right) \tag{ex-15-03.101}$$

$$= \partial_t \left(\exp\left(-4\phi\right) \right) g_{ij} + \exp\left(-4\phi\right) \partial_t g_{ij} \tag{ex-15-03.102}$$

$$= -4 \exp\left(-4\phi\right) \partial_t \phi g_{ij} + \exp\left(-4\phi\right) \partial_t g_{ij} \tag{ex-15-03.103}$$

$$= -4 \exp\left(-4\phi\right) \partial_t \phi g_{ij} - 2 \exp\left(-4\phi\right) N K_{ij} \tag{ex-15-03.104}$$

$$= \frac{2}{3} \exp\left(-4\phi\right) K N g_{ij} - 2 \exp\left(-4\phi\right) N K_{ij} \tag{ex-15-03.105}$$

$$= \frac{2}{3} \exp\left(-4\phi\right) K N g_{ij} - 2 \exp\left(-4\phi\right) N \left(A_{ij} + \frac{1}{3} g_{ij} K \right) \tag{ex-15-03.106}$$

$$= \frac{2}{3} \exp\left(-4\phi\right) K N g_{ij} - 2 \exp\left(-4\phi\right) N A_{ij} - \frac{2}{3} \exp\left(-4\phi\right) N g_{ij} K \tag{ex-15-03.107}$$

$$= -2N \exp\left(-4\phi\right) A_{ij} \tag{ex-15-03.108}$$

$$= -2N \bar{A}_{ij} \tag{ex-15-03.109}$$